7

Declarative Deponency: A Network Morphology Account of Morphological Mismatches

ANDREW HIPPISLEY

1. Introduction

IN THE FIRST CHAPTER WE WERE PRESENTED with a working definition of deponency which is based on the facts of deponent verbs in Latin (Baerman, this volume). The definition was then extended in various directions to show that morphological mismatches in languages besides Latin can be thought of as instances of 'extended' deponency. Implicit in the basic definition and its extensions is the notion of defaults and overrides. Thus if we think of Baerman's basic definition in terms of properties, the primary property is that the *default* association between form and function is *overridden*. Once the mismatch has been identified, another property is that all morphosyntactic cells in the paradigm are involved in the mismatch in the same way. This property can be viewed as a 'second order' default: having established the exceptional phenomenon, we can state its typical behaviour. But this second order default can also be overridden. For Greek there is no override: all cells are involved in the mismatch (for Greek deponents see Lavidas and Papangeli, this volume) However for Latin standard deponent verbs, a few cells consistently override this default, i.e. they are the same cells for every deponent verb (see Baerman's paper, and discussion below). Again, we can think of the consistency of the overrides in Latin as a third order default. We can then characterize semi-deponents as items that override this default by extending the set of morphosyntactic cells that do not participate in the mismatch. If these defaults concern the function, there is another (second order) default which concerns the form, namely that the deponent form is identical to the 'source' form. In Latin deponents the passive form used to realize active morphosyntax is identical to the source passive morphology used to

Proceedings of the British Academy **145**, 145–173.

realize passive morphosyntax. Thinking about it this way then allows for the possibility of an override at this level. Mismatches in Archi nominal morphology provide an example of this. We might then expect the property of defectiveness discussed in Baerman to simply fall out from whether or not an item inherits this default: where it does, homonymous forms are avoided (defectiveness), where it overrides there are distinct forms (no defectiveness). We can recast these facts about the standard case of deponency and extended deponency, using Archi deponent nouns as an example of extended deponency, as various types and levels of defaults and overrides. From this analysis four deponency properties (DPs) naturally emerge. This is shown in Table 1.

From table 1 we see that in Latin the mismatch involves voice features, whereas in Archi, our example of extended deponency, nominal number features are involved (DP1). All cells participate in the mismatch in Latin except for the present participle, the future participle, and the future infinitive; for semi-deponents even fewer cells participate in the mismatch; on the other hand, the mismatch in Archi involves all cells (DP2). Whereas Latin deponent forms are identical to the source forms, i.e. the passive form in a deponent is indistinguishable from a passive in a non-deponent, Archi behaves differently by maintaining a formal distinction between number subparadigms even in instances of deponency (DP3). Finally, Latin deponents are defective, whereas Archi deponents are not (DP4).

Corbett's account of extended deponency (this volume) implies a default situation, 'canonical inflection'; and overrides to the default, 'deviations from canonical inflection'. In this paper we take as the starting point the accounts

Table 1. Deponency properties (DPs) in terms of defaults and overrides

Default	Default level	Override	Standard dep. (Latin)	Extended dep. (Archi)	DP
function ~ form	first order	function ~ form mismatch	*overrides* voice features	*overrides* number features	DP1
function: all morphosyntactic cells involved	second order	subset of cells not involved	*overrides* 1.{pres. part. fut. infin. fut. part.} 2. more: semi-deponents	*inherits default*	DP2
form: dep. form = source form	second order	non-identity between dep. and source forms	*inherits default*	*overrides* 1. heteroclisis 2. stem shape	DP3
defective paradigm (entailed by DP3)	second order	not defective	*inherits default*	*overrides* formally distinct for number	DP4

of extended deponency by Baerman and Corbett, and aim to make explicit the role of defaults and overrides implicit in them. We do this by offering an account of Latin deponent verbs in Network Morphology, a declarative framework for writing default inheritance-based theories of the lexicon, and whose representation is underpinned by a formal semantics. We also offer an account of deponent nouns in Archi, a Nakh-Daghestanian, Lezgian language spoken in the Caucasus, as an example of extended deponency. In this way we show how the idea of extended deponency follows systematically from the starting point of a default-overrides approach to standard deponency. In section 2 we introduce the Network Morphology framework. We use part of our Network Morphology account of Latin regular verbs to illustrate the salient features of its formalism, the lexical knowledge representation language DATR (Evans and Gazdar 1996). Section 3 is then a detailed Network Morphology treatment of Latin deponent and semi-deponent verbs which recasts as a formal system of defaults and overrides the properties of deponency that Baerman draws together. Extended deponency as extension of this system is illustrated by a Network Morphology account of Archi deponent nouns (§ 4). Insights and conclusions stemming from these accounts are summarized in section 5.

2. The Formal Framework: Network Morphology

Network Morphology is a declarative framework that situates morphological facts in a network of information sharing nodes, thereby capturing the generalizations that can be made about morphology, while at the same time characterizing exceptionality. This is achieved by setting up hierarchies whose daughter nodes inherit non-monotonically, or by default, from their mothers. The broadest level of generalization stated at the dominating nodes may be overridden by intermediary nodes which themselves can express a narrower level of generalization. In this way Network Morphology provides a framework for uncovering systematicity behind prima facie irregular phenomena such as deponency, and teases apart 'true' exceptionality from semi-regularity. Network Morphology theories are expressed in DATR, a unification-based lexical knowledge representation language designed by Evans and Gazdar to support default and exceptional behaviour (Evans and Gazdar 1996). DATR defines networks by links typed by attribute paths through which information is inherited. Its interpretability comes from an explicit theory of inference, and an explicit declarative semantics. It is therefore computationally implementable, and a variety of interpreters exist for DATR theories (e.g. Cahill and Gazdar 1999: f.n. 7). For the Network

Morphology framework, see Corbett and Fraser (1993); Evans, Brown, and Corbett (2001) and Hippisley (2001) and the references there.

To illustrate, consider the small fragment of our Network Morphology theory of Latin in (1a, b). It expresses generalizations about stem shapes for a class of Latin 1st conujugation verbs. Note that the full Network Morphology theory of Latin verbs and Archi nouns is available online (Hippisley 2006a, b).

(1a)

```
CONJ_1:
    <stem theme> == ā
    <stem 1> == "<root>" <stem theme>
    <stem 2> == <stem 1> v
    <stem 3> == <stem 1> t.
```

Morphological facts are expressed as attribute paths linked to descriptors. Facts are collected at nodes to express generalizations. In (1a) we have a node which is labelled CONJ_1, which contains four facts. The first fact (second line) is that the theme vowel of this conjugation is a long /a/. This is represented by linking the attribute path <stem theme> to the atomic value ā via the inference symbol ==. The second fact is about the formation of the first stem (or 'stem 1') in this conjugation. The descriptor is not an atom, rather a series of attribute paths; this expresses that the value for <stem 1> will be the concatenation of the values of the paths <root> and <stem theme>. Descriptors can also be a combination of attribute paths and atoms, as in the cases of <stem 2> and <stem 3>. As the first conjugation is the most productive one, we can think of it as the 'default' conjugation. In (1b) we have the daughter node CONJ_4. Its hierarchical relationship with CONJ_1 is expressed by an empty path linked to the node CONJ_1, representing the (default) inheritance of all facts from that node. As DATR defines default inheritance networks, facts from the mother node may be overridden simply by specifying them. (1b) shows that CONJ_4 is identical to CONJ_1 in every regard except for the value of the theme vowel, and the addition of a fact about the first stem.

(1b)

```
CONJ_4:
    <> == CONJ_1
    <stem theme> == ī
    <stem 1 ext> == <stem 1> ē.
```

We should also note that paths as descriptors may express inheritance of a value both from the local context, i.e. found at the node where it is situated, and the global context. Global inheritance is expressed by quoting the path. In (1a) part of the value of <stem 2> is the value of a path locally declared,

<stem 1>. But the evaluation of <stem 1> depends partly on the quoted path "<root>", and this path is not available at the CONJ_1 node. Its global context is the node representing the query lexical entry. In (2) we give the example of the lexical entry for *amō* 'love', to which we give the perspicuous label Amo. Note that DATR requires that all nodes have an initial uppercase character, and that atoms must always be lower case. As the path <root> is declared at the lexical entry node, its value am can be inherited by the path <stem 1> at the CONJ_1 node.

(2)

```
Amo:
    <> == VERB
    <gloss> == love
    <root> == am
    <stem> == CONJ_1.
```

DATR also supports orthogonal multiple inheritance, i.e. values of paths can be inherited from more than one node. In (2) we see that the primary source of inheritance is from the node VERB, representing the lexical entries' word class membership, and that the conjugation class node is a secondary source of inheritance, orthogonal to the principle source. This allows for more flexibility in the distribution of morphological facts. They can belong to orthogonally related hierarchies, within a single network. (2) shows that in our Network Morphology theory of Latin we have (at least) a hierarchy for word classes, and a separate hierarchy for stem formation facts.

The final point to make concerns DATR's principle of default inference which enables it to define default inheritance networks. It is based on the fact that any path which is an extension of another path receives the same definition as the path it extends, unless otherwise stated. Put another way, by default a path implies any further specification of itself. From (1a, b) we can see that the conjugation class nodes specify an associated theme vowel. For example in CONJ_4 this is ī and for CONJ_1 it is ā. This value is inherited by a lexical entry by setting up a path <stem> that inherits from a conjugation class node. What is inherited is all paths that extend the subpath <stem>, including <stem theme>. In other words <stem> implies <stem 1>, <stem 2>, <stem 3> *and* <stem theme> (for theme vowel). This allows a lexical entry to pick up a natural class of facts collected at a node by identifying them with an attribute that is the leading subpath. The partial theorem of Amo is given in (3) and shows the value of its three stems, and its theme vowel. Note that theorems are the outputs of a DATR theory (description). Whereas theories employ a double equals (==) expressing inference, theorems are the result of making the inference, i.e. the evaluation, and employ single equals (=).

(3)

```
Amo:<gloss> = love.
Amo:<root> = am.
Amo:<stem theme> =
Amo:<stem 1> = am ā.
Amo:<stem 2> = am ā v.
Amo:<stem 3> = am ā t.
Amo:<syn active perfect past indicative sg 2> = am ā v istī.
...
```

3. Defaults and Standard Deponency: A Network Morphology Account of Latin Deponent Verbs

We briefly recall the basic facts about Latin deponent verbs that our Network Morphology account must capture. The mismatch is between morphosyntactic voice and its formal expression, specifically *active* morphosyntactic features are realized by *passive* morphology. Second, deponent verbs are defective: there are no forms available for realizing passive categories. Third, certain of the active categories are not involved in the mismatch. For some of these, this is because there is no corresponding form in the passive paradigm: a mismatch is de facto impossible. But this does not result in a gap in a deponent's paradigm since suitable active forms are used in these instances. However for one category, the future infinitive, the active is used even though there is a passive form available. Fourth, a consequence of the mismatch involves the addition of a morphosyntactic category that is generally absent because there is no morphological form to realize it. This is the active perfect past participle, realized by the passive perfect past participle form. Fifth, verbs may be semi-deponents: part of the paradigm involves a mismatch, and part is regular. Which part of the paradigm is deponent varies between verbs.

We base our account on Baerman's report of Latin deponency (Baerman 2006), who in turn draws on Ernout and Thomas (1953), Flobert (1975), and Kühner (1955). The theorem is validated by checking against the Kennedy Latin primer (Kennedy 1962). We begin by discussing our formal treatment of the verb system to capture generalizations that can be stated over regular verbs. We are then in a position to express deponent verbs as overrides to a high level default on the one hand, and as a class with it own default properties on the other.

3.1. The Latin verbal system: a Network Morphology account

We follow Aronoff's (1992) analysis of the four Latin verb conjugations as a function of one of four theme vowels available to build a Latin stem. For the realization of a given morphosyntactic category, the theme vowel is used to imply the correct combination of stem type, stem shape, and the inflectional suffix. Stem types, or indexed stems, are argued for in Aronoff (1994) and are assumed in formal theories of morphology, e.g. Network Morphology (see Hippisley 1998, 2001 and Brown 1998 for Network Morphology accounts of Russian using indexed stems) and Paradigm Function Morphology (Stump 2001).

All verbs inherit ultimately from the node VERB, shown in (4). This node holds the generalization that syntactic features by default correspond to morphological features. The DATR theory expresses how these morphosyntactic features are realized as the combination of a (global) stem path and atomic value, as we will show. The VERB node also contains the generalization that the primary division of the verbal paradigm is in terms of voice. This is expressed by the attribute ordering of paths: the first feature attribute is either active or passive. The path is then extended at active and passive formation nodes, where it is passed on appropriately for further extensions until the full path inherits a value. This allows us to capture the first property of Latin deponents, that the mismatch involves voice (Deponency Property 1, Table 1).

(4)

```
VERB:
    <syn> == "<mor>"
    <mor active> == ACT_FORMS:<>
    <mor passive> == PASS_FORMS:<>.
```

The secondary division of the paradigm is along aspectual lines. Formally this means that attributes expressing aspect features are ordered after those expressing voice features. As we shall see this enables us to capture the fact that semi-deponents are really verbs that are deponent in only one aspect (Deponency Property 2, Table 1). This is formalized in (5a, b).

(5a)

```
ACT_FORMS:
    <imperfective> == ACT_IMPF:<>
    <perfect> == ACT_PERF:<>.
```

(5b)

```
PASS_FORMS:
    <imperfective> == PASS_IMPF:<>
    <perfect> == PASS_PERF:<>.
```

The next division is according to tense; formally an attribute representing a tense feature follows the aspect attribute. This is followed by a mood attribute, then number attribute, and finally person attribute. We illustrate this ordering for the fully specified path <mor active perfective past indicative sg 2> with the nodes in (6). Note that the path <plus> in (6a) expresses the pluperfect tense feature in Latin.

(6a)

```
ACT_PERF:
    <past> == ACT_PAST_PERFECT:<>
    <future> == ACT_FUTURE_PERFECT:<>
    <plus> == ACT_PLU_PERFECT:<>.
```

(6b)

```
ACT_PAST_PERFECT:
    <indicative sg 1> == "<stem 2>" ī
    <indicative sg 2> == "<stem 2>" istī
    <indicative sg 3> == "<stem 2>" it
    <subjunctive sg 1> == "<stem 2>" erim
    <subjunctive sg 2> == "<stem 2>" erīs
    <subjunctive sg 3> == "<stem 2>" erit
    <infinitive> == "<stem 2>" isse.
```

As with all Network Morphology theories, our theory of Latin verbs takes an inferential-realizational approach (Stump 2001). The function is realized as a modification of the stem. Formally this is expressed as the path <mor active perfective past indicative sg 2> inheriting the complex value of (a) the value of the path <stem 2> and (b) the atom istī. We also assume that a lexical entry can have more than one modifiable stem. In this case what is being modified is the value of a path labelled 'stem 2' that is retrievable from the lexical entry being queried; this is what is expressed by the double quotes (see discussion about global inheritance in the previous section). There are (typically) three stems which are used to realize the full set of morphosyntactic features, and the shape of the three stems is generalizable. We capture this using nodes to hold the generalizations and then allowing lexical entries to inherit from them. There is an association between stem shape and conjugation class, and to preserve this association we label the stem formation nodes after the four conjugation classes used in traditional analyses of Latin verbs. These are shown in (7), and were partially discussed in section 2.

(7a)

```
CONJ_1:
    <stem theme> == ā
    <stem 1> == "<root>""<stem theme>"
    <stem 2> == "<stem 1>" v
    <stem 3> == "<stem 1>" t.
```

(7b)

```
CONJ_2:
    <> == CONJ_1
    <stem theme> == ē
    <stem 2> == "<root>" u
    <stem 3> == "<root>" it.
```

(7c)

```
CONJ_3:
    <> == CONJ_1
    <stem theme> == e
    <stem 2> == "<root>" s
    <stem 3> == "<root>" t
    <stem 1 alt> == "<root>" i.
```

(7d)

```
CONJ_4:
    <> == CONJ_1
    <stem theme> == ī
    <stem 1 ext> == <stem 1> ē.
```

From (7a–d) we can see that each conjugation class node also specifies an associated theme vowel. For example in Conjugation 2 (7b) this is /ē/. The value of the theme vowel for a lexical entry is expressed as the value of a path that extends <stem>, and is used for determining the shape of stem 1 (see section 2). The partial theorem of Amo is given in (8) and shows the value of its three stems, and the realization of the morphosyntactic category 'active perfective past indicative 2nd person singular'.

(8)

```
Amo:<gloss> = love.
Amo:<root> = am.
Amo:<stem 1> = am ā.
Amo:<stem 2> = am ā v.
Amo:<stem 3> = am ā t.
Amo:<syn active perfect past indicative sg 2> = am ā v istī.
. . .
```

We follow Aronoff (1992) in using the theme vowel, the value of the path <stem theme>, not only to build stems but also to determine stem type and inflectional marker combinations when they differ amongst conjugations for a given feature. For example, for the active imperfective future, one set of desinences is attached to Stem 1 for Conjugations 1 and 2, and another set of desinences is attached directly to the root for Conjugation 3 and to Stem 1 for Conjugation 4. This is shown in the partial theorems of four verbs each belonging to one of the four Conjugations. (9a) is a Conjugation 1 verb, (9b) belongs to Conjugation 2, (9c) Conjugation 3, and (9d) Conjugation 4.

(9a)

Amo:<gloss> = love.
Amo:<root> = am.
Amo:<stem 1> = am ā.
Amo:<syn active imperfective future indicative sg 1> = am ā bō.
Amo:<syn active imperfective future indicative sg 2> = am ā bis.
Amo:<syn active imperfective future indicative sg 3> = am ā bit.
. . .

(9b)

Moneo:<gloss> = advise.
Moneo:<root> = mon.
Moneo:<stem 1> = mon ē.
Moneo:<syn active imperfective future indicative sg 1> = mon ē bō.
Moneo:<syn active imperfective future indicative sg 2> = mon ē bis.
Moneo:<syn active imperfective future indicative sg 3> = mon ē bit.
. . .

(9c)

Rego:<gloss> = rule.
Rego:<root> = reg.
Rego:<syn active imperfective future indicative sg 1> = reg am.
Rego:<syn active imperfective future indicative sg 2> = reg ēs.
Rego:<syn active imperfective future indicative sg 3> = reg et.
. . .

(9d)

Audio:<gloss> = hear.
Audio:<root> = aud.
Audio:<stem 1> = aud ī.
Audio:<syn active imperfective future indicative sg 1> = aud ī am.
Audio:<syn active imperfective future indicative sg 2> = aud ī ēs.
Audio:<syn active imperfective future indicative sg 3> = aud ī et.
. . .

The theory for this is given in (10) where (10a) show that evaluation is based on the theme vowel of the lexical entry being queried: <"<stem theme>">. It should be made clear that for the realization of most categories there is no conjugation distinction if a stem indexing approach is taken. In other words, the conjugational distinctions in the formation of the various stem types allow us to make a general statement about stem type and desinence as the spell-out for most categories, which is inherited by default by (regular) verbs of any class. The way we express the realization of 'active perfective past indicative 2nd person singular' in (5) and (6) above is how we are able to handle most of the categories.

(10a)

```
ACT_IMPF_FUT:
    <indicative> == ACT_IMPF_FUT_INDIC:<"<stem theme>">
    <infinitive> == "<mor active imperfective future participle>" esse
    <participle> == "<stem 3>" ūrus.
```

(10b)

```
ACT_IMPF_FUT_INDIC:
    <ā> == TYPE_1_ACT_FUT_INDIC:<>
    <e> == TYPE_2_ACT_FUT_INDIC:<>
    <ī> == TYPE_3_ACT_FUT_INDIC:<>
    <ē> == <ā>.
```

(10c)

```
TYPE_1_ACT_FUT_INDIC:
    <sg 1> == "<stem 1>" bō
    <sg 2> == "<stem 1>" bis
    <sg 3> == "<stem 1>" bit.
```

(10d)

```
TYPE_2_ACT_FUT_INDIC:
    <sg 1> == "<root>" am
    <sg 2> == "<root>" ēs
    <sg 3> == "<root>" et.
```

(10e)

```
TYPE_3_ACT_FUT_INDIC:
    <sg 1> == "<stem 1>" am
    <sg 2> == "<stem 1>" ēs
    <sg 3> == "<stem 1>" et.
```

To end this preliminary section on our theory of the Latin verbal system, we should briefly note that where possible we capture cases of directional syncretism as path referrals, in the spirit of Network Morphology (see Baerman, Brown, and Corbett 2005: ch. 5). For example, in (10a) the active imperfective future infinitive is partially realized by the active imperfective future participle. In (11a) we show how we capture the fact that for all verbs the active future perfect indicative is syncretic with the active perfect past subjunctive. (11c) shows the identities in form in the theorem for the Conjugation 2 verb *moneō* 'advise'.

(11a)

```
ACT_FUTURE_PERFECT:
    <indicative sg 1> == "<stem 2>" erō
```

```
    <indicative sg 2> == “<mor active perfect past subjunctive sg 2>”
    <indicative sg 3> == “<mor active perfect past subjunctive sg 3>”.
```

(11b)

```
ACT_PERF:
    <past> == ACT_PAST_PERFECT:<>
    . . .
ACT_PAST_PERFECT:
    <subjunctive sg 1> == “<stem 2>” erim
    <subjunctive sg 2> == “<stem 2>” erīs
    <subjunctive sg 3> == “<stem 2>” erit
    . . .
```

(11c)

```
Moneo:<gloss> = advise.
Moneo:<root> = mon.
Moneo:<stem 1> = mon ē.
Moneo:<stem 2> = mon u.
Moneo:<stem 3> = mon it.
Moneo:<syn active perfect past subjunctive sg 1> = mon u erim.
Moneo:<syn active perfect past subjunctive sg 2> = mon u erīs.
Moneo:<syn active perfect past subjunctive sg 3> = mon u erit.
Moneo:<syn active perfect future indicative sg 1> = mon u erō.
Moneo:<syn active perfect future indicative sg 2> = mon u erīs.
Moneo:<syn active perfect future indicative sg 3> = mon u erit.
. . .
```

Having laid out the main points of our Network Morphology account of the verb system, we can now turn to the defining characteristics of deponent verbs and express these formally as the properties of standard deponency as outlined in Table 1 in section 1.

3.2. Deponency Property 1: overriding the function ~ form association

The primary characteristic of Latin deponent verbs is that they realize active morphosyntax with passive morphology. In terms of defaults and overrides, they override a first order default that active morphosyntax is realized by active morphology. This default has been formalized in (4) and (5a) above. Deponency Property 1 (DP1) is expressed as an equation at a deponency node from which deponent verbs inherit (12a).

(12a)

```
DEPONENT:
    <> == VERB
    <mor active> == PASS_FORMS:<>
    . . .
```

(12b)

```
PASS_FORMS:
    <imperfective> == PASS_IMPF:<>
    <perfect> == PASS_PERF:<>.
```

As deponency only involves a single feature, the voice feature, and not aspect, tense, mood, number, or person, and because in our theory of Latin verbs we have partitioned the paradigm according to voice, we can express deponency parsimoniously by referencing the path <mor active> referring to the node which handles passive morphology. This is because formally <mor active> is the leading subpath that implies all its extensions, e.g. <mor active perfect past indicative sg 2>, and hence gathers together the full set of active morphological features. All these fully specified paths are then evaluated at passive morphology nodes: from (12b) we see that an active imperfective will be evaluated at a passive imperfective node, and an active perfect at a passive perfective node. Note that by positing a special deponency node, we are capturing deponency as a set of second level defaults: the node overrides the broadest level default of function ~ form associations, but contains its own generalizations for the class of deponent verbs. In (13a) we have the lexical entry for the deponent verb *hortor* 'encourage'.

(13a)

```
Hortor:
    <> == DEPONENT
    <gloss> == encourage
    <root> == hort
    <stem> == CONJ_1.
```

Thus if we compare (13a) with the lexical entry for *amō* 'love' in (2) we see that the only difference between a lexical entry for a deponent verb and a regular verb is the main source of inheritance. Deponent verbs, just as regular verbs, have stem types and theme vowels which are specified by their conjugation class; they also inherit their own set of default facts. The partial theorem of Hortor is given in (13b), and this can be compared to the partial theorem of Amo in (13c). (Note that the lines in bold face will be discussed in § 3.3).

(13b)

```
Hortor:<gloss> = encourage.
Hortor:<root> = hort.
Hortor:<stem 1> = hort ā.
Hortor:<stem 2> = hort ā v.
Hortor:<stem 3> = hort ā t.
Hortor:<syn active imperfective present indicative sg 2> = hort ā ris.
Hortor:<syn active imperfective present indicative sg 3> = hort ā tur.
```

Hortor:<syn active imperfective present infinitive> = hort ā rī.
Hortor:<syn active imperfective present participle> = hort ā ns.
Hortor:<syn active imperfective future infinitive> = hort ā t ūrus esse.
Hortor:<syn active imperfective future participle> = hort ā t ūrus.
Hortor:<syn active perfect past indicative sg 2> = hort ā t us es.
Hortor:<syn active perfect past indicative sg 3> = hort ā t us est.
Hortor:<syn active perfect past participle> = hort ā t us.
. . .

(13c)

Amo:<gloss> = love.
Amo:<root> = am.
Amo:<stem 1> = am ā.
Amo:<stem 2> = am ā v.
Amo:<stem 3> = am ā t.
Amo:<syn active imperfective present indicative sg 2> = am ā s.
Amo:<syn active imperfective present indicative sg 3> = am ā t.
Amo:<syn active imperfective present infinitive> = am ā re.
Amo:<syn active imperfective present participle> = am ā ns.
Amo:<syn active imperfective future infinitive> = am ā t ūrus esse.
Amo:<syn active imperfective future participle> = am ā t ūrus.
Amo:<syn active perfect past indicative sg 2> = am ā v istī.
Amo:<syn active perfect past indicative sg 3> = am ā v it.
. . .

The theorem clearly shows the mismatch between syntactic active function and passive morphology. Recall from the introduction to section 2 that Network Morphology theories aim to uncover systematicity behind irregular phenomena such as deponency. DP1 is expressed as an override of a default, but this override is situated at a node form which lexical entries such as the one for *hortor* inherit. In this way can express deponency itself as having default properties for a class of items.

3.3. Deponency Property 2: cells involved in the misatch

Once we have established that there is a mismatch (DP1) we could assume that all cells of the paradigm are involved in the mismatch, and view this as a second order default: a default about the behaviour of the exceptional phenomenon. The inheritance or overriding of this default characterizes Deponency Property 2. In standard deponency the default is overridden: three morphosyntactic categories do not participate in the mismatch. These can be clearly seen when we compare the theorem of deponent verb with a regular verb, as in (13b) and (13c). The categories in question are highlighted in boldface. For two of these, the imperfective present participle and the imperfective future participle, the mismatch is impossible because there are no passive equivalents in a verb's paradigm. Deponent verbs then default to active morphology for these categories, thereby avoiding gaps in the paradigm. This is captured by defaulting to active formation unless otherwise specified, and is represented in (14a) for the case of present participle (last equation), and in (14b) for the future participle.

(14a)

```
PASS_IMPF_PRES:
    <indicative sg 2> == "<stem 1>" ris
    <indicative sg 3> == "<stem 1 alt>" tur
    <subjunctive> == PASS_PRES_SUBJ:<"<stem theme>">
    <imperative sg 2> == VERB:<mor active imperfective present infinitive>
    <imperative pl 2> == "<stem 1 alt>"minī
    <infinitive> == PASS_PRES_INF:<"<stem theme>">
    <> == ACT_IMPF_PRES.
```

(14b)

```
PASS_IMPF_FUT:
    <indicative> == PASS_FUT_INDIC:<"<stem theme>">
    <infinitive> == "<stem 3>" um īrī
    <> == ACT_IMPF_FUT.
```

The empty path in the last lines of (14a) and (14b) expresses any extension of the subpath not defined at the node; this is referred to an equivalent active formation node for its value. In (14b) this will be the (future) participle as the indicative and infinitive are defined, and there are no active future imperative or subjunctive categories. And in (14a) this will be the (present) participle as it is the only feature not defined at the node. However, (14b) shows quite clearly that there is in Latin a passive future infinitive form, and we see it in the theorem of Amo in (14c). Despite its availability, for deponent verb Hortor the active form is used, *hortātūrus esse* (compare Amo's theorem). In this case we have a category not involved in the mismatch when the equivalent passive form is available. This is a stipulation we must therefore make for deponent verbs, and is expressed at the Deponent node, a more complete version of which is now given in (15).

(15)

```
DEPONENT:
    <> == VERB
    <mor active> == PASS_FORMS:<>
    <mor active imperfective future infinitive> == VERB
    <syn active perfect past participle> ==
        VERB:<mor passive perfect past participle>
    . . .
```

We could think of this as a 'true' overriding of the second order default, whereas the other two categories override by virtue of the inheritance of another default, i.e. use active morphology in the absence of passive morphology. We noted in section 1 that DP2 is different for Greek in that it inherits the default that the mismatch will involve all cells. This can now be explained in terms of the availability of mediopassive morphology: the same set of active features is present for the mediopassive. Alongside this fact about Greek regular morphology is the fact that the deponent verbs do not stipulate non-participation in the mismatch of a particular category. This means that the statement <mor active> == MEDIO_PASSIVE_FORMS:<> would suffice to account for the mismatch in Greek.

While there are two categories in the active paradigm which are missing in the passive paradigm, there is one category in the passive missing from the active. This is the perfect past participle. This means that as a consequence of a mismatch with passive morphology, deponent verbs are able to realize a category that cannot be realized for regular verbs, namely the active past participle. Latin has a form for passive past participle, which is used. This is expressed as a syntactic function in (15), where the leading subpath is <syn>, inheriting from a passive form, where the leading subpath is <mor>. Since this is the only place where the extension of this <syn> path is specified, the absence of morphology for this function for non-deponent verbs results in it not appearing in their theorems.

3.3.1. Deponency Property 2 and semi-deponent verbs

For semi-deponent verbs it is not just a few cells that are not involved in the mismatch, but half the paradigm. Semi-deponents can therefore be seen as one extreme of Deponency Property 2. The division of the paradigm could be seen as being one of aspect. Another view is to see the division based on Stem 2 forms, but as they only involve realization of perfect categories (in the active), it makes no additional claim: in either case the perfect is involved. This is different to Archi, for example, where the deponency is a function of the stem types as we shall see in section 4. To illustrate we show the theory for the verb *audeō* 'dare', which is deponent for perfect morphosyntax, and regular for imperfective features, and the verb *revertor* 'return' which is the converse: it is deponent for imperfective morphosyntactic features, and regular for perfective. (16a) and (16b) give the nodes that express generalizations about these two types of semi-deponent verb. It should be noted that semi-deponents traditionally refer to what I call Perfect Deponent, and what I call Imperfect Deponent is discussed as allied phenomena.

(16a)

```
PERFECT_DEPONENT:
    <> == DEPONENT
    <mor active imperfective> == VERB.
```

(16b)

```
IMPF_DEPONENT:
    <> == DEPONENT
    <mor active perfect> == VERB.
```

The node in (16a) expresses that deponency properties are inherited by default, except that the active imperfective forms override the generalization that their values are those of passive forms; instead they behave like any other verb. (16b) is the converse: this time active perfect forms override the mismatching generalization stated at the Deponent node. This analysis makes the claim that semi-deponency is a type of deponent verb, with some overrides. The lexical entry for perfect deponent *audeō* 'dare' and its partial theorem is given in (17a, b).

(17a)

```
Audeo:
    <> == PERFECT_DEPONENT
    <gloss> == dare
    <root> == aud
    <stem 3> == aus
    <stem> == CONJ_2.
```

(17b)

```
Audeo:<gloss> = dare.
Audeo:<root> = aud.
Audeo:<stem 1> = aud ē.
Audeo:<stem 2> = aud u.
Audeo:<stem 3> = aus.
Audeo:<syn active imperfective present indicative sg 2> = aud ē s.
Audeo:<syn active imperfective present indicative sg 3> = aud ē t.
Audeo:<syn active imperfective present infinitive> = aud ē re.
Audeo:<syn active imperfective present participle> = aud ē ns.
Audeo:<syn active imperfective future infinitive> = aus ūrus esse.
Audeo:<syn active imperfective future participle> = aus ūrus.
Audeo:<syn active perfect past indicative sg 2> = aus us es.
Audeo:<syn active perfect past indicative sg 3> = aus us est.
Audeo:<syn active perfect past participle> = aus us.
. . .
```

For the imperfect deponent verb *revertor*, only the imperfective cells are involved in the mismatch. At the same time, we have seen that a higher level default for deponent verbs states that three imperfective categories are not involved (Deponency Property 3). As *revertor* is expressed as a type of deponent verb, it inherits this default with the result that it has even fewer deponent forms than a semi-deponent verb like *audeō*. In other words, it displays an extreme version of the Deponency Property 3 default, that all cells are involved in the mismatch, by overriding it for the majority of cells. In addition, Revertor inherits a syntactic active perfect past participle, spelled out as a passive form: *reversus*. This is possible despite Revertor's perfect subparadigm not being involved in the mismatch because realization of a syntactic active perfect past participle is a fact stated at the node Deponency.

(18a)

```
Revertor:
    <> == IMPF_DEPONENT
    <gloss> == return
    <root> == revert
    <stem 2> == <root>
    <stem 3> == <root> s
    <stem> == CONJ_3.
```

(18b)

```
    Revertor:<gloss> = return.
    Revertor:<root> = revert.
    Revertor:<stem 1> = revert e.
    Revertor:<stem 2> = revert.
    Revertor:<stem 3> = revert s.
    Revertor:<syn active imperfective present indicative sg 2> = revert e ris.
    Revertor:<syn active imperfective present indicative sg 3> = revert i tur.
    Revertor:<syn active imperfective present infinitive> = revert ī.
    Revertor:<syn active imperfective present participle> = revert e ns.
    Revertor:<syn active imperfective future infinitive> = revert s ūrus
      esse.
    Revertor:<syn active imperfective future participle> = revert s ūrus.
    Revertor:<syn active perfect past indicative sg 1> = revert ī.
    Revertor:<syn active perfect past indicative sg 2> = revert istī.
    Revertor:<syn active perfect past indicative sg 3> = revert it.
    Revertor:<syn active perfect past participle> = revert s us.
. . .
```

3.4 Deponency Properties 3 and 4: defectiveness

In our theory defectiveness is captured by adding a fact at the deponency node, as shown in (19). Here the path <mor passive> implies all its extensions, the full set of passive features, and is assigned the value undefined.

(19)

```
DEPONENT:
    <> == VERB
    <mor active> == PASS_FORMS:<>
    <mor active imperfective future infinitive> == VERB
```

<syn active perfect past participle> == VERB:<mor passive perfect past participle>
<mor passive> == undefined.

As indicated in Table 1, however, defectiveness should really be seen as a natural consequence of inheriting another default. Once we have established that there is a mismatch in function and form, we can make generalizations about the function, outlined as Deponency Property 2 in Table 1, and generalizations about the form, outlined as Deponency Property 3 (Table 1). The form type of generalization is that the deponent form is non-distinct from the source form. This can be stated as a second order default, and in standard deponency (Latin) it is not overridden. This default is bound in with the way the first order default, namely that active inherits passive morphology, is expressed (12a, b). We can see this by comparing the deponent active forms for *hortor* in (13b) with the passive forms of the regular verb *amō* in (20)

(20)

Amo:<gloss> = love.
Amo:<root> = am.
Amo:<stem 1> = am ā.
Amo:<stem 2> = am ā v.
Amo:<stem 3> = am ā t.
Amo:<syn passive imperfective present indicative sg 2> = am ā ris.
Amo:<syn passive imperfective present indicative sg 3> = am ā tur.
Amo:<syn passive imperfective present infinitive> = am ā rī.
Amo:<syn passive imperfective future infinitive> = am ā t um īrī.
Amo:<syn passive perfect past indicative sg 2> = am ā t us es.
Amo:<syn passive perfect past indicative sg 3> = am ā t us est.
Amo:<syn passive perfect past participle> = am ā t us.
. . .

4. Defaults and Extended Deponency: A Network Morphology Account of Archi Deponent Nouns

Archi deponent nouns involve the mismatch between morphosyntactic number and its formal expression. This is an example of extended deponency because it displays Deponency Property 1, i.e. that the default function ~ form association is being overridden, but this override extends beyond voice features. It also displays Deponency Property 2, but again in the 'extended' sense: it inherits the second level default that all morphosyntactic cells are involved in the mismatch, unlike Latin. For Deponency Property 4, rather than inheriting the default of defective paradigms it overrides with non-defectiveness. This is entailed by displaying Deponency Property 3 in the extended sense: unlike Latin there is a distinction between the deponent forms, used to express one number, and the source forms, used to express the other number. The distinction is by virtue of heteroclisis. But it is also due to the fact that Archi nouns may have a version of the root associated with a particular number, seen in cases of weak and strong suppletion. In this event, there is no mismatch between morphosyntactic number and its formal expression at the level of the root, even in a deponent noun.

In this section we present our Network Morphology account of the Archi noun system as an example of a formal defaults-overrides account of extended deponency. The analysis of the Archi data is taken from Baerman (2006b) which in turn draws on Kibrik (1977a; 1977b).

4.1 The Archi nominal system: a Network Morphology account

Archi nouns have distinct singular and plural stems on which are built the case inflectional markers. Number distinctions are therefore expressed by the stem. Oblique case markers are based on the ergative word-form. There are no inflectional classes as such: all nouns that can inflect for a morphosyntactic feature inflect in the same way. However there are three main stem formation classes that provide suitable number distinguishing stems for inflection. Regardless of class, by default the absolute singular form is based directly on the root. Where a noun has a distinct root for all plural forms, i.e. in weak and strong suppleting nouns, the absolutive singular is formed on the singular root. For some nouns there is a distinct root used for the absolutive singular. These facts about Archi are formalized in (21) to (24) below. In (21) we express the default dependence of a morphological word-form on the lexeme's stem, and in this way formalize number marking through the stem. This is because the path <mor> implies any extension of itself, as does the path <stem>. This expresses the fact that, for example, the path <mor sg> takes as its value the value of the path <stem sg>. The implication is that all singular word forms, regardless of case, will begin with a singular stem. The quotes indicate global inheritance expressing that <stem> and its extensions, <stem sg> and <stem pl> and as we shall see later <stem sg erg>, are retrieved from the lexical entry being queried. Finally, the 'exception' is the absolutive singular which is formed directly on a root form: in the case of some items, on the bare root, for others a special form of the root used in the singular only, and for others a special root used for the absolutive singular only. These three possibilities correspond to the subpath <root>, its extension <root sg> and its further extension <root sg abs>. Again, the quotes indicate that the path is lexically specified.

(21)

```
NOUN:
    <mor> == "<stem>"
    <mor sg abs> == "<root sg abs>".
```

With respect to the formation of the absolutive and ergative, Archi can be said to have three major nouns classes. These generalize the formation of the stem when it combines with inflections to express the full set of morphosyntactic features. One class is for lexemes with consonant final roots, which we call Class 1. Another is for lexemes with vowel final roots (Class 2). There is also a special class for substantivized adjectives, and these lexemes have frozen gender markers (Class 3). The three stem formation classes are given in (22) and (23) below. In (22a) the first class, Class 1, inherits facts about nouns, including those discussed above, and specifies singular and plural stem building generalizations. The singular stem is based on the lexically specified ergative root, in those instances where this is distinct, and the plural on a plural root, again in instances where there is a distinction in the roots, to which is added the formative -*mul*-.

(22a)

```
CLASS_1:
    <> == NOUN
    <stem sg> == "<root sg erg>" CLASS_1_SG:<>
    <stem pl> == "<root pl>" mul CLASS_1_PL:<>.
```

For singular stem formation, referral is made to a special node (22b) where by default the formative -*li*- is added to the root, and then case endings are picked up at a separate node (discussed below). However, for the absolutive nothing is added to the root, and no reference is made to inflectional material to mark case ending.

(22b)

```
CLASS_1_SG:
    <> == li NOUN_FORMS:<>
    <abs> == .
```

Plural stem formation is expressed similarly with the important difference that the formative -*čaj*- is added to all stems; the exception is the absolutive marker which is zero (21c).

(22c)

```
CLASS_1_PL:
    <> == čaj NOUN_FORMS:<>
    <abs> == .
```

Singular and plural stem formation operates differently in the other two classes, and this is shown in (23). Note that Class 2 operates in the same was as Class 1 for singular stem formation, and this is expressed by using the empty path.

(23)

```
CLASS_2:
    <> == CLASS_1
    <stem pl> == "<root pl>" CLASS_2_PL:<>.
CLASS_2_PL:
    <> == t:aj NOUN_FORMS:<>
    <abs> == t:u.
CLASS_3:
    <> == NOUN
    <stem sg> == "<root sg erg>" CLASS_3_SG:<>
    <stem pl> == "<root pl>" CLASS_3_PL:<>.
CLASS_3_SG:
    <> == mu NOUN_FORMS:<>.
CLASS_3_PL:
    <> == maj NOUN_FORMS:<>
    <abs> == .
```

Each stem is ultimately referred to a list of case features and corresponding inflections, shown in (24). As word-forms are based on the ergative word-form, the ergative receives no additional marker.

(24)

```
NOUN_FORMS:
    <erg> ==
    <gen> == n
    <dat> == s
    <comit> == ɬ:u
    <comp> == xur
    <perm> == kɬ'əna
    <part> == qˤiš
    <superlat> == t:ik
    <sublat> == kɬ'ak.
```

Having provided a formal account of the noun system, we can now show how lexical entries inherit these facts. Lexical entries are represented as nodes which are labelled to indicate the lexeme being represented. Each lexical entry is furnished with a root. For some, a special version of the root is used for building the oblique singular stems and another for building plural stems. This is true for *Kɬ'ánnu* 'lover'. Lexical entries inherit from one of the three stem formation classes. This is shown in (25a–c).[1]

[1] Note that accented characters as title-case in nodes names cannot be expressed in DATR and are used here purely for exposition

(25a)

```
Árum:
    <> == CLASS_1
    <gloss> == sickle
    <root> == áʕrum .
```

(25b)

```
Aʕri:
    <> == CLASS_2
    <gloss> == military division
    <root> == áʕri.
```

(25c)

```
Kɬ’ánnu:
    <> == CLASS_3
    <gloss> == lover
    <root> == kɬ’ánnu
    <root sg erg> == <root> m
    <root pl> == kɬ’ánnib.
```

4.2 Extended deponency in Archi

The mismatch between morphosyntactic feature and formal expression involves number. The formal part of the mismatch lies at the level of the stem. In other words the mismatch is between stem shape and the number feature normally identified with it. The word *xali* ‘family’ is a deponent noun whose singular morphosyntax is realised by what appear to be plural forms. So the dative singular is *xal-majs* which is formally similar to Class 3 plural nouns, such as *kɬ’ánnib-majs* (DAT. PL.) ‘lovers’. This is clear evidence that the first level default is being overridden, the function ~ form association. As it does not involve voice features, it is an extended version of DP1. At the same time, every cell in the paradigm is involved in the mismatch. In other words, it inherits the default associated with DP2, and so displays an extended version of DP2. An interesting feature of Archi deponency is that there are no defective paradigms, as with Latin. While a noun may have a deponent plural form with singular meaning, it also has a full subparadigm of plural forms with plural meaning. This is shown in the theorem of the lexical entry for *xali* generated by our Network Morphology theory of Archi..[2]

(26)

```
Xali:<gloss> = family.
Xali:<mor sg abs> = xali.
```

[2] The theory does not take into account the regular /aj/ ~ /e/ alternation before a consonant.

```
Xali:<mor sg erg> = xal maj.
Xali:<mor sg gen> = xal maj n.
Xali:<mor sg dat> = xal maj s.
Xali:<mor sg comit> = xal maj ɬ:u.
Xali:<mor sg comp> = xal maj xur.
Xali:<mor sg perm> = xal maj kɬ'əna.
Xali:<mor sg part> = xal maj qˤiš.
Xali:<mor sg superlat> = xal maj t:ik.
Xali:<mor sg sublat> = xal maj kɬ'ak.
Xali:<mor pl abs> = xali t:u.
Xali:<mor pl erg> = xali t:aj.
Xali:<mor pl gen> = xali t:aj n.
Xali:<mor pl dat> = xali t:aj s.
Xali:<mor pl comit> = xali t:aj ɬ:u.
Xali:<mor pl comp> = xali t:aj xur.
Xali:<mor pl perm> = xali t:aj kɬ'əna.
Xali:<mor pl part> = xali t:aj qˤiš.
Xali:<mor pl superlat> = xali t:aj t:ik.
Xali:<mor pl sublat> = xali t:aj kɬ'ak.
```

This is possible because the two sub-paradigms belong to different stem formation classes. In other words, in the case of Deponency Property 3, Archi overrides the default that the deponent forms in a lexem'es paradigm are non-distinct from the source forms in the same paradigm *by virtue* of heteroclisis. The Network Morphology account of the deponent lexeme is given in (27). The lexeme is a Class 3 noun the singular, and is treated a Class 2 noun for the plural (last two lines). We should also note that this lexical item has a distinct ergative singular root on which all singular forms apart from absolutive singular are built. Examples where a distinct root is used for the entire singular subparadigm are discussed in the next section.

(27)

```
Xali:
    <> == NOUN
    <gloss> == family
    <root> == xali
    <root sg erg> == xal
    <stem sg> == <root sg erg> CLASS_3_PL:<>
    <stem pl> == <root pl> CLASS_2_PL:<>.
```

4.3. Number marking maintained in the mismatch

We have seen how heteroclisis in deponent nouns is responsible for overriding the default that source and deponent forms are non-distinct. The subparadigms of deponent nouns are distinguished for another reason. In Archi some nouns have more than one root form, i.e. forms which are not generalizable by the stem formation classes. In other words these items are examples of suppleting lexemes. In such cases one version of the root is used for singular realizations, and the other for plural realizations. This is important for our next example of deponency in Archi, and to help illustrate it we can use the non-deponent suppletive noun for 'corner of a sack': *bič'ni* (ABS SG.) and *boždo* (ABS PL) (see Hippisley et al. 2004 for a fuller discussion of suppletion and paradigmatic organisation, including this Archi example). We express the distinction at the level of the lexeme, as shown in (28a) which is the lexical entry for *bič'ni* 'corner of a sack'. There is one form for the singular, the 'singular root', and one for the plural, 'plural root'. The theorem is shown in (28b).[3]

(28a)

```
Bič'ni:
    <> == CLASS_1
    <gloss> == corner of sack
    <root sg> == bič'ni
    <root pl> == boždo
    <stem pl> == <root pl> CLASS_1_PL:<>.
```

(28b)

```
Bič'ni:<gloss> = corner of sack.
Bič'ni:<mor sg abs> = bič'ni.
Bič'ni:<mor sg erg> = bič'ni li.
Bič'ni:<mor sg gen> = bič'ni li n.
Bič'ni:<mor sg dat> = bič'ni li s.
Bič'ni:<mor sg comit> = bič'ni li ɬ:u.
Bič'ni:<mor sg comp> = bič'ni li xur.
Bič'ni:<mor sg perm> = bič'ni li kɬ'əna.
Bič'ni:<mor sg part> = bič'ni li qˤiš.
Bič'ni:<mor sg superlat> = bič'ni li t:ik.
Bič'ni:<mor sg sublat> = bič'ni li kɬ'ak.
Bič'ni:<mor pl abs> = boždo.
Bič'ni:<mor pl erg> = boždo čaj.
Bič'ni:<mor pl gen> = boždo čaj n.
Bič'ni:<mor pl dat> = boždo čaj s.
Bič'ni:<mor pl comit> = boždo čaj ɬ:u.
Bič'ni:<mor pl comp> = boždo čaj xur.
Bič'ni:<mor pl perm> = boždo čaj kɬ'əna.
Bič'ni:<mor pl part> = boždo čaj qˤiš.
```

[3] In the plural there should be an epenthetic /r/ between the root final vowel and the initial consonant of the *-čaj-* formative, e.g. *boždorčaj* (PL, ERG) but this allomorphy is not lexical and is therefore not accounted for by the theory.

Bič'ni:<mor pl superlat> = boždo čaj t:ik.
Bič'ni:<mor pl sublat> = boždo čaj kɬ'ak.

There is one deponent noun which also has suppletive roots. The noun *xˁon* 'cow' has the suppletive root *buc:'i* used in realizations of plural morphosyntax. The lexical entry for this noun is given in (29a) with its theorem in (29b).

(29a)

Xˁon:
<> == NOUN
<gloss> == cow
<root sg abs> == xˁon
<root sg erg> == xˁini
<root pl> == buc:'i
<stem sg> == <root sg erg> NOUN_FORMS:<>
<stem pl> == <root pl> CLASS_1_SG:<>.

(29b)

Xˁon:<gloss> = cow.
Xˁon:<mor sg abs> = xˁon.
Xˁon:<mor sg erg> = xˁini.
Xˁon:<mor sg gen> = xˁini n.
Xˁon:<mor sg dat> = xˁini s.
Xˁon:<mor sg comit> = xˁini ɬ:u.
Xˁon:<mor sg comp> = xˁini xur.
Xˁon:<mor sg perm> = xˁini kɬ'əna.
Xˁon:<mor sg part> = xˁini qˁiš.
Xˁon:<mor sg superlat> = xˁini t:ik.
Xˁon:<mor sg sublat> = xˁini kɬ'ak.
Xˁon:<mor pl abs> = buc:'i.
Xˁon:<mor pl erg> = buc:'i li.
Xˁon:<mor pl gen> = buc:'i li n.
Xˁon:<mor pl dat> = buc:'i li s.
Xˁon:<mor pl comit> = buc:'i li ɬ:u.
Xˁon:<mor pl comp> = buc:'i li xur.
Xˁon:<mor pl perm> = buc:'i li kɬ'əna.
Xˁon:<mor pl part> = buc:'i li qˁiš.
Xˁon:<mor pl superlat> = buc:'i li t:ik.
Xˁon:<mor pl sublat> = buc:'i li kɬ'ak.

The default associated with Deponency Property 3 rules out such examples: source and deponent forms are non-distinct, therefore suppletion based on the function involved in the mismatch is not possible. In Latin a verb cannot be both deponent and suppletive. As this default is overridden in Archi anyway due to heteroclisis, with the result that there are formally distinct sub-paradigms, suppletion is maintained even in a deponent nouns. We should note that for *xˁon* 'cow' the heteroclisis involves the singular sub-paradigm not belonging to any of the three stem formation classes, and inheriting directly from a node generalizing over all nouns, but the plural belonging to Class 1. In sum, the extended version of Deponency Property 3 allows for deponent items to be suppletive. Another example of a deponent noun which is suppletive is *c'aj* 'goat' which has the suppletive root *c'ohor* for the plural. The lexical entry is given in (30), and expresses its suppletion, heteroclisis, and deponency behaviours.

(29)

```
C'aj:
    <> == NOUN
    <gloss> == goat
    <root> == c'aj
    <root pl> == c'ohor
    <stem sg> == <root sg erg> CLASS_2_PL:<>
    <stem pl> == <root pl> CLASS_1_PL:<>.
```

5. Concluding Remarks

We have attempted to make explicit the role that defaults and overrides play in the notion 'extended deponency'. We have done this by recasting the characteristics of standard and extended deponency as sets of defaults and overrides operating at different levels of generalization, which we have then formalized within the declarative defaults-based framework of Network Morphology. We have presented Network Morphology accounts of standard deponency (Latin) and extended deponency (Archi). At the broadest level, deponency is a mismatch of function and form. We have expressed deponency as the overriding of a default operating at the highest level of generalization. We can then state two generalizations about deponent items: first, they involve all cells in the paradigm, and second within a deponent lexeme's paradigm the deponent forms are indistinguishable from the source forms. These two properties are expressed as second level defaults, defaults about deponent items themselves. Standard deponency and extended deponency can be characterized in the ways in which these secondary level defaults are inherited or overridden, and if overridden, the extent to which they are overridden. Semi-deponency is directly connected with the first of these second level defaults, that all cells participate in the mismatch, and is where an item overrides the default to the extent that half its paradigm is not involved in the mismatch. And defectiveness is directly connected to the second default: inheriting it leads to potential mass inflectional homonymy between two sub-paradigms, so one of the sub-paradigms is 'laid aside'. In our example of extended deponency, this default is overridden due to heteroclisis and the function (number) being expressed by the root. The result is the use of the deponent form for its source function. In this way we have shown how the definition of extended deponency is stateable in terms of defaults and overrides, allowing for the possibility of formal, computationally tractable accounts of deponency phenomena.

References

Aronoff, Mark. 1992. Stems in Latin verbal morphology. *Morphology now*, ed. by Mark Aronoff, 5–32. SUNY: New York.

—— 1994. *Morphology* by itself: stems and inflectional classes. Cambridge, Massachusetts: MIT Press.

Baerman, Matthew. 2006a. Latin deponency language report. [Available at www.smg.surrey.ac.uk/deponency/Examples/Latin.htm].

—— 2006b. Archi deponency language Report [Available at www.smg.surrey.ac.uk/deponency/Examples/Archi.htm]

—— This volume. Morphological typology of deponency.

Baerman, Matthew, Brown, Dunstan, and Greville G. Corbett. 2005. *The syntax-morphology interface: a study of syncretism.* Cambridge: CUP.

Brown, Dunstan. 1998. Stem indexing and morphonological selection in the Russian verb. *Models of inflection*, ed. by Ray Fabri, Albert Ortmann, and Teresa Parodi, 196–221. Niemeyer: Tübingen.

Cahill, Lynn, and Gerald Gazdar. 1999. German noun inflection. *Journal of Linguistics* 35.1–42.

Corbett, Greville G. This volume. Deponency, syncretism, and what lies between.

Corbett, Greville G., and Norman Fraser. 1993. Network Morphology: a DATR account of Russian nominal inflection. *Journal of Linguistics* 29.113–42.

Ernout Alfred, and François Thomas. 1953. *Syntaxe latine* (2nd edn). Paris: Klincksieck.

Evans, Nicholas, Dunstan Brown, and Greville G. Corbett. 2001. Dalabon pronominal prefixes and the typology of syncretism. *Yearbook of Morphology 2000*, ed. by Jaap van Marle and Geert Booij, 187–231. Dordrecht: Kluwer:.

Evans, Roger, and Gerald Gazdar. 1996. DATR: a language for lexical knowledge representation. *Computational Linguistics* 22.167–216.

Flobert Pierre. 1975. *Les verbes déponents latins des origines à Charlemagne*. Paris: Société d'Édition 'Les Belles Lettres'.

Hippisley Andrew. 1998. Indexed stems and Russian word formation: a Network Morphology account of Russian personal nouns. *Linguistics* 36.1039–1124.

—— 2001. Word Formation Rules in a default inheritance framework: a Network Account of Russian personal nouns. *Yearbook* of Morphology 1999, ed. by Jaap van Marle and Geert Booij, 221–61. Dordrecht: Kluwer.

—— 2006a. A formal account of Latin deponent verbs. [Available at www.smg.surrey.ac.uk/deponency/DATR_analyses/Archi/Latin_report.pdf]

—— 2006b. A formal account of Archi deponent nouns. [Available at www.smg.surrey.ac.uk/deponency/DATR_analyses/Archi/Archi_report.pdf]

Hippisley, Andrew, Marina Chumakina, Greville G. Corbett, and Dunstan Brown. 2004. Suppletion: frequency, categories and distribution of stems. *Studies in Language* 28.389–421.

Kennedy, Benjamin. 1962. *Revised Latin primer* (edited and further revised by J. Mountford). Harlow: Longman.

Kibirik, Aleksandr E. 1977a. *Opyt strukturnogo opisanija arčinskogo jazyka:* II: *Taksonomičeskaja grammatika.* (Publikacii otdelenija strukturnoj i prikladnoj lingvistiki 12). Moscow: Izdatel´stvo Moskovskogo universiteta

Kibirik, Aleksandr E. 1977b. *Arčinskij* jazyk: teksty i slovar´. Moscow: Izdatel´stvo Moskovskogo Universiteta.

Kühner, Raphael. 1955. *Ausführliche Grammatik der lateinischen Sprache* (part 2, 1, revised by Andreas Thierfelder). Hamburg: Hahn.

Stump, Gregory. 2001. *Inflectional morphology: a theory of paradigm structure*. Cambridge: CUP.